Rule 1.2.1: Int[(a+b x+c x^2)^p, x] → Int121[a, b, c, p, x]

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_.,x_Symbol] := Int121[a,b,c,p,x] /;
  FreeQ[{a,b,c,p},x]
```

```
Int121::usage =
  "Int121[a,b,c,p,x] returns the antiderivative of (a+b x+c x^2)^p wrt x.";
```

```
Int121[a_,b_,c_,p_,x_] :=
  If[EqQ[p,0],
    (a+b*x+c*x^2)^p*x,
  If[EqQ[c,0],
    Int111[a,b,p,x],
  If[EqQ[b,0],
    Int151[a,c,p,2,x],
  If[EqQ[b^2-4*a*c,0],
    If[IntegerQ[p],
      (b/2+c*x)^(2*p+1)/(c^(p+1)*(2*p+1)),
    (a+b*x+c*x^2)^FracPart[p]/(c^IntPart[p]*(b/2+c*x)^(2*FracPart[p])) * Int111[b/2,c,2*p,x]],
  If[IntegerQ[p],
    If[EqQ[p,1],
      a*x + b*x^2/2 + c*x^3/3,
    If[EqQ[a,0],
      Int[Apart[x^p*(b+c*x)^p,x],x],
    If[EqQ[p,-1],
      If[NiceSqrtQ[b^2-4*a*c],
        With[{q=Rt[b^2-4*a*c,2]}, 2*c/q * Int111[b-q,2*c,-1,x] - 2*c/q * Int111[b+q,2*c,-1,x]],
      With[{q=1-4*Simplify[a*c/b^2]}, If[RationalQ[q] && (EqQ[q^2,1] || Not[RationalQ[b^2-4*a*c]]),
        -2/b * Subst[Int151[q,-1,-1,2,x],x,1+2*c*x/b],
      -2 * Subst[Int151[Simplify[b^2-4*a*c],-1,-1,2,x],x,b+2*c*x]]]],
    If[NiceSqrtQ[b^2-4*a*c] && Not[FractionalPowerFactorQ[Rt[b^2-4*a*c,2]]],
      With[{q=Rt[b^2-4*a*c,2]}, 1/c^p * Int[Apart[(b/2-q/2+c*x)^p*(b/2+q/2+c*x)^p,x],x]],
    If[GtQ[p,0],
      Int[Apart[(a+b*x+c*x^2)^p,x],x],
    (b+2*c*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) - 2*c*(2*p+3)/((p+1)*(b^2-4*a*c)) * Int121[a,b,c,p+1,x]]]]]],
  If[GtQ[p,0] && (IntegerQ[4*p] || IntegerQ[3*p]),
    (b+2*c*x)*(a+b*x+c*x^2)^p/(2*c*(2*p+1)) - p*(b^2-4*a*c)/(2*c*(2*p+1)) * Int121[a,b,c,p-1,x],
  If[LtQ[p,-1] && (IntegerQ[4*p] || IntegerQ[3*p]),
    If[EqQ[p,-3/2],
      -2*(b+2*c*x)/((b^2-4*a*c)*Sqrt[a+b*x+c*x^2]),
    (b+2*c*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) - 2*c*(2*p+3)/((p+1)*(b^2-4*a*c)) * Int121[a,b,c,p+1,x]],
  If[EqQ[a,0],
    If[LtQ[b^2/c,0],
      1/(2^(2*p+1)*c*(-c/(b^2))^p) * Subst[Int151[1,-1/b^2,p,2,x],x,b+2*c*x],
    If[EqQ[p,-1/2],
      2 * Subst[Int151[1,-c,-1,2,x],x,x/Sqrt[b*x+c*x^2]],
```

```
    If[IntegerQ[4*p] || IntegerQ[3*p],
      (b*x+c*x^2)^p/(-c*(b*x+c*x^2)/(b^2))^p * Int121[0,-c/b,-c^2/b^2,p,x],
    -(b*x+c*x^2)^(p+1)/(b*(p+1)*(-c*x/b)^(p+1))*Hypergeometric2F1[-p,p+1,p+2,1+c*x/b]]]],
  If[LtQ[c/Simplify[b^2-4*a*c],0],
    With[{q=Simplify[b^2-4*a*c]}, 1/(2^(2*p+1)*c*(-c/q)^p) * Subst[Int151[1,-1/q,p,2,x],x,b+2*c*x]],
  If[EqQ[p,-1/2],
    2 * Subst[Int151[4*c,-1,-1,2,x],x,(b+2*c*x)/Sqrt[a+b*x+c*x^2]],
  If[IntegerQ[4*p] || IntegerQ[3*p],
    With[{k=Denominator[p]}, k*Sqrt[(b+2*c*x)^2]/(b+2*c*x) * Subst[Int152[1,k*(p+1)-1,b^2-4*a*c,4*c,-1/2,k,x],x,(a+b*x+c*x^2)^(1/
  With[{q=Rt[b^2-4*a*c,2]}, -(a+b*x+c*x^2)^(p+1)/(q*(p+1)*((q-b-2*c*x)/(2*q))^(p+1))*Hypergeometric2F1[-p,p+1,p+2,(b+q+2*c*x)/(2*
```