Rules for integrands of the form $\left(a+b\,x+c\,x^2\right)^p$

- **Case when $b == 0$ handled by rules for integrands of the form $(a+b\,x^n)^p$.**

1. $\int\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0$

1: $\int\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$

- **Derivation: Algebraic simplification and power rule for integration**
- **Basis: If $b^2-4\,a\,c==0$, then $a+b\,x+c\,x^2==\frac{1}{c}\left(\frac{b}{2}+c\,x\right)^2$**
- **Rule 1.2.1.1.1: If $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$, then**

$$\int\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{1}{c^p}\int\left(\frac{b}{2}+c\,x\right)^{2\,p}\,dx\,\longrightarrow\,\frac{\left(\frac{b}{2}+c\,x\right)^{2\,p+1}}{c^{p+1}\,(2\,p+1)}$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  (b/2+c*x)^(2*p+1)/(c^(p+1)*(2*p+1)) /;
FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis: If $b^2-4\,a\,c==0$, then $\partial_x\frac{(a+b\,x+c\,x^2)^p}{\left(\frac{b}{2}+c\,x\right)^{2\,p}}==0$**
- **Rule 1.2.1.1.2: If $b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$, then**

$$\int\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{\left(a+b\,x+c\,x^2\right)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]}\left(\frac{b}{2}+c\,x\right)^{2\,\text{FracPart}[p]}}\int\left(\frac{b}{2}+c\,x\right)^{2\,p}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/(c^IntPart[p]*(b/2+c*x)^(2*FracPart[p])) * Int[(b/2+c*x)^(2*p),x] /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int (a + b\,x + c\,x^2)^p \, dx$ when $p \in \mathbb{Z}$

1: $\int (b\,x + c\,x^2)^p \, dx$ when $p \in \mathbb{Z}$

- **Derivation: Algebraic expansion**
- **Rule 1.2.1.2.1: If $p \in \mathbb{Z}$, then**

$$\int (b\,x + c\,x^2)^p \, dx \longrightarrow \int \text{ExpandIntegrand}[x^p\,(b + c\,x)^p,\,x]\, dx$$

- **Program code:**

```
Int[(b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[x^p*(b+c*x)^p,x],x] /;
FreeQ[{b,c},x] && IntegerQ[p]
```

2. $\int \frac{1}{a + b\,x + c\,x^2} \, dx$

1: $\int \frac{1}{a + b\,x + c\,x^2} \, dx$ when NiceSqrtQ$[b^2 - 4\,a\,c]$

- **Derivation: Algebraic expansion**
- **Basis: Let $q \to \sqrt{b^2 - 4\,a\,c}$, then** $\frac{1}{a+b\,x+c\,x^2} == \frac{2\,c}{q\,(b-q+2\,c\,x)} - \frac{2\,c}{q\,(b+q+2\,c\,x)}$
- **Rule 1.2.1.2.2.1: If NiceSqrtQ$[b^2 - 4\,a\,c]$, let $q \to \sqrt{b^2 - 4\,a\,c}$, then**

$$\int \frac{1}{a + b\,x + c\,x^2} \, dx \longrightarrow \frac{2\,c}{q} \int \frac{1}{b - q + 2\,c\,x} \, dx - \frac{2\,c}{q} \int \frac{1}{b + q + 2\,c\,x} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q * Int[1/(b-q+2*c*x),x] - 2*c/q * Int[1/(b+q+2*c*x),x]] /;
FreeQ[{a,b,c},x] && NiceSqrtQ[b^2-4*a*c]
```

2: $\int \frac{1}{a+b\,x+c\,x^2}\,dx$ when $b^2-4\,a\,c\notin\mathbb{R}\,\bigwedge\,1-\frac{4\,a\,c}{b^2}\in\mathbb{R}$

- **Reference: G&R 2.172.4, CRC 109, A&S 3.3.16**
- **Reference: G&R 2.172.2, CRC 110a, A&S 3.3.17**
- **Derivation: Integration by substitution**
- **Basis: Let** $q\to 1-\frac{4\,a\,c}{b^2}$**, then** $\frac{1}{a+b\,x+c\,x^2}=-\frac{2}{b}\,\mathrm{Subst}\left[\frac{1}{q-x^2},\,x,\,1+\frac{2\,c\,x}{b}\right]\partial_x\left(1+\frac{2\,c\,x}{b}\right)$
- **Rule 1.2.1.2.2.2: If** $b^2-4\,a\,c\notin\mathbb{R}$**, let** $q\to 1-\frac{4\,a\,c}{b^2}$**, if** $q\in\mathbb{R}\,\bigwedge\,\left(q^2==1\,\bigvee\,b^2-4\,a\,c\notin\mathbb{R}\right)$**, then**

$$\int \frac{1}{a+b\,x+c\,x^2}\,dx\,\longrightarrow\,-\frac{2}{b}\,\mathrm{Subst}\left[\int\frac{1}{q-x^2}\,dx,\,x,\,1+\frac{2\,c\,x}{b}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  With[{q=1-4*Simplify[a*c/b^2]},
  -2/b * Subst[Int[1/(q-x^2),x],x,1+2*c*x/b] /;
 RationalQ[q] && (EqQ[q^2,1] || Not[RationalQ[b^2-4*a*c]])] /;
FreeQ[{a,b,c},x]
```

3: $\int \frac{1}{a+b\,x+c\,x^2}\,dx$

- **Reference: G&R 2.172.2, CRC 110a, A&S 3.3.17**
- **Reference: G&R 2.172.4, CRC 109, A&S 3.3.16**
- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{a+b\,x+c\,x^2}=-2\,\mathrm{Subst}\left[\frac{1}{b^2-4\,a\,c-x^2},\,x,\,b+2\,c\,x\right]\partial_x\,(b+2\,c\,x)$
- **Rule 1.2.1.2.2.3:**

$$\int \frac{1}{a+b\,x+c\,x^2}\,dx\,\longrightarrow\,-2\,\mathrm{Subst}\left[\int\frac{1}{b^2-4\,a\,c-x^2}\,dx,\,x,\,b+2\,c\,x\right]$$

- **Program code:**

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  -2 * Subst[Int[1/(Simplify[b^2-4*a*c]-x^2),x],x,b+2*c*x] /;
FreeQ[{a,b,c},x]
```

3: $\int \left(a+b\,x+c\,x^2\right)^p\,dx$ when $p \in \mathbb{Z}\ \land$ NiceSqrtQ$\left[b^2-4\,a\,c\right]$

- **Derivation: Algebraic expansion**
- **Basis: Let** $q \to \sqrt{b^2-4\,a\,c}$ **, then** $a+b\,x+c\,x^2 == \frac{1}{c}\left(\frac{b}{2}-\frac{q}{2}+c\,x\right)\left(\frac{b}{2}+\frac{q}{2}+c\,x\right)$
- **Rule 1.2.1.2.3: If** $p \in \mathbb{Z}\ \land$ NiceSqrtQ$\left[b^2-4\,a\,c\right]$**, let** $q \to \sqrt{b^2-4\,a\,c}$ **, then**

$$\int \left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow \frac{1}{c^p}\int \text{ExpandIntegrand}\left[\left(\frac{b}{2}-\frac{q}{2}+c\,x\right)^p\left(\frac{b}{2}+\frac{q}{2}+c\,x\right)^p,\ x\right]dx$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  1/c^p * Int[ExpandIntegrand[(b/2-q/2+c*x)^p*(b/2+q/2+c*x)^p,x],x] /;
 Not[FractionalPowerFactorQ[q]]] /;
FreeQ[{a,b,c},x] && IntegerQ[p] && NiceSqrtQ[b^2-4*a*c]
```

4: $\int \left(a+b\,x+c\,x^2\right)^p\,dx$ when $p \in \mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule 1.2.1.2.4: If** $p \in \mathbb{Z}^+$**, then**

$$\int \left(a+b\,x+c\,x^2\right)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[\left(a+b\,x+c\,x^2\right)^p,\ x\right]dx$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[p,0]
```

5: $\int \left(a+b\,x+c\,x^2\right)^p\,dx$ when $p+1 \in \mathbb{Z}^-$

- **Reference: G&R 2.171.3, G&R 2.263.3, CRC 113, CRC 241**
- **Derivation: Quadratic recurrence 2a with** $m = 0$**,** $A = 1$ **and** $B = 0$
- **Rule 1.2.1.2.5: If** $p+1 \in \mathbb{Z}^-$**, then**

$$\int \left(a+b\,x+c\,x^2\right)^p\,dx \;\longrightarrow\; \frac{(b+2\,c\,x)\,\left(a+b\,x+c\,x^2\right)^{p+1}}{(p+1)\,\left(b^2-4\,a\,c\right)}-\frac{2\,c\,(2\,p+3)}{(p+1)\,\left(b^2-4\,a\,c\right)}\int \left(a+b\,x+c\,x^2\right)^{p+1}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) -
  2*c*(2*p+3)/((p+1)*(b^2-4*a*c)) * Int[(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c},x] && ILtQ[p,-1]
```

3: $\int \left(a+b\,x+c\,x^2\right)^p\,dx$ when $p>0 \land (4\,p\in\mathbb{Z} \lor 3\,p\in\mathbb{Z})$

- **Reference: G&R 2.260.2, CRC 245, A&S 3.3.37**
- **Derivation: Quadratic recurrence 1b with** $m=-1$, $A=d$ **and** $B=e$
- **Rule 1.2.1.3: If** $p>0 \land (4\,p\in\mathbb{Z} \lor 3\,p\in\mathbb{Z})$, **then**

$$\int \left(a+b\,x+c\,x^2\right)^p\,dx \;\longrightarrow\; \frac{(b+2\,c\,x)\,\left(a+b\,x+c\,x^2\right)^{p}}{2\,c\,(2\,p+1)}-\frac{p\,\left(b^2-4\,a\,c\right)}{2\,c\,(2\,p+1)}\int \left(a+b\,x+c\,x^2\right)^{p-1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^p/(2*c*(2*p+1)) -
  p*(b^2-4*a*c)/(2*c*(2*p+1)) * Int[(a+b*x+c*x^2)^(p-1),x] /;
FreeQ[{a,b,c},x] && GtQ[p,0] && (IntegerQ[4*p] || IntegerQ[3*p])
```

4. $\int (a + b x + c x^2)^p \, dx$ when $p < -1 \land (4 p \in \mathbb{Z} \lor 3 p \in \mathbb{Z})$

1: $\int \frac{1}{(a + b x + c x^2)^{3/2}} \, dx$ when $b^2 - 4 a c \neq 0$

- **Reference: G&R 2.264.5, CRC 239**
- **Derivation: Quadratic recurrence 2a with $m = 0$, $A = 1$, $B = 0$ and $p = -\frac{3}{2}$**
- **Rule 1.2.1.4.1: If $b^2 - 4 a c \neq 0$, then**

$$\int \frac{1}{(a + b x + c x^2)^{3/2}} \, dx \longrightarrow -\frac{2 (b + 2 c x)}{(b^2 - 4 a c) \sqrt{a + b x + c x^2}}$$

- **Program code:**

```
Int[1/(a_.+b_.*x_+c_.*x_^2)^(3/2),x_Symbol] :=
  -2*(b+2*c*x)/((b^2-4*a*c)*Sqrt[a+b*x+c*x^2]) /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

2: $\int (a + b x + c x^2)^p \, dx$ when $p < -1 \land (4 p \in \mathbb{Z} \lor 3 p \in \mathbb{Z})$

- **Reference: G&R 2.171.3, G&R 2.263.3, CRC 113, CRC 241**
- **Derivation: Quadratic recurrence 2a with $m = 0$, $A = 1$ and $B = 0$**
- **Rule 1.2.1.4.2: If $p < -1 \land (4 p \in \mathbb{Z} \lor 3 p \in \mathbb{Z})$, then**

$$\int (a + b x + c x^2)^p \, dx \longrightarrow \frac{(b + 2 c x) (a + b x + c x^2)^{p+1}}{(p + 1) (b^2 - 4 a c)} - \frac{2 c (2 p + 3)}{(p + 1) (b^2 - 4 a c)} \int (a + b x + c x^2)^{p+1} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) -
  2*c*(2*p+3)/((p+1)*(b^2-4*a*c)) * Int[(a+b*x+c*x^2)^(p+1),x] /;
FreeQ[{a,b,c},x] && LtQ[p,-1] && (IntegerQ[4*p] || IntegerQ[3*p])
```

5. $\int \left(b\,x + c\,x^2\right)^p \,dx$

1: $\int \left(b\,x + c\,x^2\right)^p \,dx$ when $\frac{b^2}{c} < 0$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{b^2}{c} < 0$, **then** $\left(b\,x + c\,x^2\right)^p == \frac{1}{2^{2\,p+1}\,c\,\left(-\frac{c}{b^2}\right)^p}\,\text{Subst}\left[\left(1 - \frac{x^2}{b^2}\right)^p,\ x,\ b + 2\,c\,x\right]\,\partial_x\,(b + 2\,c\,x)$
- **Rule 1.2.1.5.1: If** $\frac{b^2}{c} < 0$, **then**

$$\int \left(b\,x + c\,x^2\right)^p \,dx \longrightarrow \frac{1}{2^{2\,p+1}\,c\,\left(-\frac{c}{b^2}\right)^p}\,\text{Subst}\left[\int \left(1 - \frac{x^2}{b^2}\right)^p \,dx,\ x,\ b + 2\,c\,x\right]$$

- **Program code:**

```
Int[(b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  1/(2^(2*p+1)*c*(-c/(b^2))^p) * Subst[Int[(1-x^2/b^2)^p,x],x,b+2*c*x] /;
FreeQ[{b,c,p},x] && LtQ[b^2/c,0]
```

2: $\int \frac{1}{\sqrt{b\,x + c\,x^2}} \,dx$

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\sqrt{b\,x+c\,x^2}} == 2\,\text{Subst}\left[\frac{1}{1-c\,x^2},\ x,\ \frac{x}{\sqrt{b\,x+c\,x^2}}\right]\,\partial_x\,\frac{x}{\sqrt{b\,x+c\,x^2}}$
- **Rule 1.2.1.5.2:**

$$\int \frac{1}{\sqrt{b\,x + c\,x^2}} \,dx \longrightarrow 2\,\text{Subst}\left[\int \frac{1}{1 - c\,x^2} \,dx,\ x,\ \frac{x}{\sqrt{b\,x + c\,x^2}}\right]$$

- **Program code:**

```
Int[1/Sqrt[b_.*x_+c_.*x_^2],x_Symbol] :=
  2 * Subst[Int[1/(1-c*x^2),x],x,x/Sqrt[b*x+c*x^2]] /;
FreeQ[{b,c},x]
```

3: $\int (b\,x + c\,x^2)^p\,dx$ when $4\,p \in \mathbb{Z} \lor 3\,p \in \mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(b\,x+c\,x^2)^p}{\left(-\frac{c\,(b\,x+c\,x^2)}{b^2}\right)^p} == 0$
- **Note: Resulting integrand is of the form** $(b\,x + c\,x^2)^p$ **where** $\frac{b^2}{c} < 0$**.**
- **Note: If this optional rule is deleted, the resulting antiderivative is less compact but real when the integrand is real.**
- **Rule 1.2.1.5.3: If** $4\,p \in \mathbb{Z} \lor 3\,p \in \mathbb{Z}$**, then**

$$\int (b\,x + c\,x^2)^p\,dx \longrightarrow \frac{(b\,x + c\,x^2)^p}{\left(-\frac{c\,(b\,x+c\,x^2)}{b^2}\right)^p} \int \left(-\frac{c\,x}{b} - \frac{c^2\,x^2}{b^2}\right)^p dx$$

- **Program code:**

```
Int[(b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  (b*x+c*x^2)^p/(-c*(b*x+c*x^2)/(b^2))^p * Int[(-c*x/b-c^2*x^2/b^2)^p,x] /;
FreeQ[{b,c},x] && (IntegerQ[4*p] || IntegerQ[3*p])
```

4: $\int \left(b\,x+c\,x^2\right)^p dx$ when $4\,p\notin\mathbb{Z}\,\bigwedge\,3\,p\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{\left(b\,x+c\,x^2\right)^p}{x^p\,(b+c\,x)^p} = 0$
- **Rule 1.2.1.5.4: If** $4\,p\notin\mathbb{Z}\,\bigwedge\,3\,p\notin\mathbb{Z}$**, then**

$$\int \left(b\,x+c\,x^2\right)^p dx$$

$$\longrightarrow$$

$$\frac{\left(b\,x+c\,x^2\right)^p}{x^p\,(b+c\,x)^p}\int x^p\,(b+c\,x)^p\,dx$$

$$\longrightarrow$$

$$-\frac{\left(b\,x+c\,x^2\right)^{p+1}}{b\,(p+1)\,\left(-\frac{c\,x}{b}\right)^{p+1}}\,\text{Hypergeometric2F1}\left[-p,\;p+1,\;p+2,\;1+\frac{c\,x}{b}\right]$$

- **Program code:**

```
Int[(b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -(b*x+c*x^2)^(p+1)/(b*(p+1)*(-c*x/b)^(p+1))*Hypergeometric2F1[-p,p+1,p+2,1+c*x/b] /;
FreeQ[{b,c,p},x] && Not[IntegerQ[4*p]] && Not[IntegerQ[3*p]]
```

6: $\int \left(a+b\,x+c\,x^2\right)^p\,dx$ when $\frac{c}{b^2-4\,a\,c}<0$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{c}{b^2-4\,a\,c}<0$, **then** $\left(a+b\,x+c\,x^2\right)^p=\frac{1}{2^{2\,p+1}\,c\,\left(-\frac{c}{b^2-4\,a\,c}\right)^p}\,\text{Subst}\left[\left(1-\frac{x^2}{b^2-4\,a\,c}\right)^p,\,x,\,b+2\,c\,x\right]\,\partial_x\,(b+2\,c\,x)$
- **Rule 1.2.1.6: Let** $q\to b^2-4\,a\,c$, **if** $\frac{c}{q}<0$, **then**

$$\int \left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow\;\frac{1}{2^{2\,p+1}\,c\,\left(-\frac{c}{q}\right)^p}\,\text{Subst}\left[\int\left(1-\frac{x^2}{q}\right)^p\,dx,\,x,\,b+2\,c\,x\right]$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Simplify[b^2-4*a*c]},
  1/(2^(2*p+1)*c*(-c/q)^p) * Subst[Int[(1-x^2/q)^p,x],x,b+2*c*x] /;
 LtQ[c/q,0]] /;
FreeQ[{a,b,c,p},x]
```

7: $\int \frac{1}{\sqrt{a+b\,x+c\,x^2}}\,dx$

- **Reference: G&R 2.261.1, CRC 237a, A&S 3.3.33**
- **Reference: CRC 238**
- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\sqrt{a+b\,x+c\,x^2}}=2\,\text{Subst}\left[\frac{1}{4\,c-x^2},\,x,\,\frac{b+2\,c\,x}{\sqrt{a+b\,x+c\,x^2}}\right]\,\partial_x\,\frac{b+2\,c\,x}{\sqrt{a+b\,x+c\,x^2}}$
- **Rule 1.2.1.7:**

$$\int \frac{1}{\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;2\,\text{Subst}\left[\int\frac{1}{4\,c-x^2}\,dx,\,x,\,\frac{b+2\,c\,x}{\sqrt{a+b\,x+c\,x^2}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_+c_.*x_^2],x_Symbol] :=
  2 * Subst[Int[1/(4*c-x^2),x],x,(b+2*c*x)/Sqrt[a+b*x+c*x^2]] /;
FreeQ[{a,b,c},x]
```

8: $\int \left(a+b\,x+c\,x^2\right)^p\,dx$ when $4\,p\in\mathbb{Z}\,\vee\,3\,p\in\mathbb{Z}$

- **Derivation: Integration by substitution and piecewise constant extraction**
- **Basis: If** $k\in\mathbb{Z}^+$**, then** $\left(a+b\,x+c\,x^2\right)^p == \frac{k\sqrt{(b+2\,c\,x)^2}}{b+2\,c\,x}\,\text{Subst}\left[\frac{x^{k\,(p+1)-1}}{\sqrt{b^2-4\,a\,c+4\,c\,x^k}},\,x,\,\left(a+b\,x+c\,x^2\right)^{1/k}\right]\partial_x\left(a+b\,x+c\,x^2\right)^{1/k}$
- **Basis:** $\partial_x\frac{\sqrt{(b+2\,c\,x)^2}}{b+2\,c\,x} == 0$
- **Note: Antiderivative of resulting integral can be expressed in terms of elliptic integral functions.**
- **Rule 1.2.1.8: If** $4\,p\in\mathbb{Z}\,\vee\,3\,p\in\mathbb{Z}$**, let** $k\to\text{Denominator}[p]$**, then**

$$\int\left(a+b\,x+c\,x^2\right)^p\,dx\,\longrightarrow\,\frac{k\sqrt{(b+2\,c\,x)^2}}{b+2\,c\,x}\,\text{Subst}\left[\int\frac{x^{k\,(p+1)-1}}{\sqrt{b^2-4\,a\,c+4\,c\,x^k}}\,dx,\,x,\,\left(a+b\,x+c\,x^2\right)^{1/k}\right]$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{k=Denominator[p]},
  k*Sqrt[(b+2*c*x)^2]/(b+2*c*x) * Subst[Int[x^(k*(p+1)-1)/Sqrt[b^2-4*a*c+4*c*x^k],x],x,(a+b*x+c*x^2)^(1/k)]] /;
FreeQ[{a,b,c},x] && (IntegerQ[4*p] || IntegerQ[3*p])
```

9: $\int \left(a+b\,x+c\,x^2\right)^p\,dx$ when $4\,p\notin\mathbb{Z}\,\wedge\,3\,p\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis: Let** $q\to\sqrt{b^2-4\,a\,c}$ **, then** $\partial_x\frac{(a+b\,x+c\,x^2)^p}{(b+q+2\,c\,x)^p\,(b-q+2\,c\,x)^p}==0$
- **Rule 1.2.1.9: If** $4\,p\notin\mathbb{Z}\,\wedge\,3\,p\notin\mathbb{Z}$**, let** $q\to\sqrt{b^2-4\,a\,c}$ **, then**

$$\int \left(a+b\,x+c\,x^2\right)^p\,dx$$
$$\longrightarrow$$
$$\frac{\left(a+b\,x+c\,x^2\right)^p}{(b+q+2\,c\,x)^p\,(b-q+2\,c\,x)^p}\int (b+q+2\,c\,x)^p\,(b-q+2\,c\,x)^p\,dx$$
$$\longrightarrow$$
$$-\frac{\left(a+b\,x+c\,x^2\right)^{p+1}}{q\,(p+1)\,\left(\frac{q-b-2\,c\,x}{2\,q}\right)^{p+1}}\,\text{Hypergeometric2F1}\left[-p,\,p+1,\,p+2,\,\frac{b+q+2\,c\,x}{2\,q}\right]$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  -(a+b*x+c*x^2)^(p+1)/(q*(p+1)*((q-b-2*c*x)/(2*q))^(p+1))*Hypergeometric2F1[-p,p+1,p+2,(b+q+2*c*x)/(2*q)]] /;
FreeQ[{a,b,c,p},x] && Not[IntegerQ[4*p]] && Not[IntegerQ[3*p]]
```